

See discussions, stats, and author profiles for this publication at: <https://www.researchgate.net/publication/221440263>

Algebra of Programming Using Dependent Types

Conference Paper · July 2008

DOI: 10.1007/978-3-540-70594-9_15 · Source: DBLP

CITATIONS

9

READS

17

3 authors, including:



[Patrik Jansson](#)

Chalmers University of Technology

61 PUBLICATIONS 1,210 CITATIONS

SEE PROFILE

All content following this page was uploaded by [Patrik Jansson](#) on 16 January 2017.

The user has requested enhancement of the downloaded file. All in-text references [underlined in blue](#) are added to the original document and are linked to publications on ResearchGate, letting you access and read them immediately.

Algebra of Programming using Dependent Types

Shin-Cheng Mu¹, Hsiang-Shang Ko², and Patrik Jansson³

¹ Institute of Information Science, Academia Sinica, Taiwan

² Department of Computer Science and Information Engineering
National Taiwan University, Taiwan

³ Department of Computer Science and Engineering
Chalmers University of Technology & University of Gothenburg, Sweden

Abstract. Dependent type theory is rich enough to express that a program satisfies an input/output relational specification, but it could be hard to construct the proof term. On the other hand, squiggolists know very well how to show that one relation is included in another by algebraic reasoning. We demonstrate how to encode functional and relational derivations in a dependently typed programming language. A program is coupled with an algebraic derivation from a specification, whose correctness is guaranteed by the type system.

1 Introduction

Program derivation is the technique of successively applying formal rules to a specification to obtain a program that is correct by construction. On the other hand, modern programming languages deploy expressive type systems to guarantee compiler-verifiable properties. There has been a trend to explore the expressiveness of dependent types, which opens a whole new world of type-level programming techniques. As Altenkirch et al. [1] put it, dependently typed programs are, “by their nature, proof carrying code.” This paper aims to illustrate their comment by showing, in the dependently typed language Agda [17], that programs can actually carry their derivations.

As a teaser, Fig. 1 shows a derivation of a sorting algorithm in progress. The type of *sort-der* is a proposition that there exists a program of type $[Val] \rightarrow [Val]$ that is contained in *ordered?* \circ *permute*, a relation mapping a list to one of its ordered permutations. The proof proceeds by derivation from the specification towards the algorithm. The first step exploits monotonicity of \circ and that *permute* can be expressed as a fold. The second step makes use of relational fold fusion. The shaded areas denote *interaction points* — fragments of (proof) code to be completed. The programmer can query Agda for the expected type and the context of the shaded expression. When the proof is completed, an algorithm *isort* is obtained by extracting the witness of the proposition. It is an executable program that is backed by the type system to meet the specification.

We have developed a library for functional and relational program derivation, with convenient notation for algebraic reasoning. Our work aims to be a co-operation between the *squiggolists* and dependently-typed programmers that

```

sort-der :  $\exists ([Val] \rightarrow [Val]) (\backslash f \rightarrow ordered? \circ permute \sqsupseteq fun\ f)$ 
sort-der = exists _ (
   $ordered? \circ permute$ 
   $\sqsupseteq \langle (\backslash vs \rightarrow \text{-monotonic } ordered? (permute\text{-is-fold } vs)) \rangle$ 
   $ordered? \circ foldR\ combine\ nil$ 
   $\sqsupseteq \langle \{ foldR\text{-fusion } ordered? ins\text{-step } ins\text{-base} \} 0$ 
   $\{ \} 1 \rangle$ 
)

isort :  $[Val] \rightarrow [Val]$ 
isort = witness sort-der

```

Fig. 1. A derivation of a sorting algorithm in progress (see Sect. 4 for the details).

may benefit both sides. On the one hand, a number of tools for program transformation [22, 11, 24] have been developed but few of them have been put into much use. Being able to express derivation *within* the programming language encourages its use and serves as documentation. This paper is a case study of using the Curry-Howard isomorphism which the squiggolists may appreciate: specification of the program is expressed in their types, whose proofs (derivations) are given as programs and checked by the type system. On the other hand, it is known among dependently-typed programmers that the expressiveness of dependent types is far beyond proving that *reverse* preserves the length of its input. We can reason about the full input/output specification, for example, that *fast-reverse* is pointwise equal to the quadratic-time *reverse*, or that insertion sort implements a relational specification of sort. The reason this is rarely done is probably because it appears difficult to construct the proof terms. The method we propose is to develop the proof by algebraic reasoning within Agda.

In Sect. 2 we give an introduction to the part of Agda we use. We present our encoding of relations and their operations in Sect. 3, which prepares us to discuss our primary example in Sect. 4 and conclude with related work in Sect. 5.

2 A Crash Course on Agda

By “Agda” we mean Agda version 2, a dependently typed programming language evolved from the theorem prover having the same name. In this section we give a crash course on Agda, focusing on the aspects we need. For a detailed documentation, the reader is referred to Norell [17] and the Agda wiki [21].

Agda has a Haskell-like syntax extended with a number of additional features. Dependent function types are written $(x : A) \rightarrow B$ where B may refer to the identifier x , while non-dependent functions are written $A \rightarrow B$. The identity function, for example, can be defined by:

```

id :  $(A : Set) \rightarrow A \rightarrow A$ 
id  $A\ a = a$ ,

```

where *Set* is the kind of types. To apply *id* we should supply both the type and the value parameters, e.g., *id* \mathbb{N} 3 where \mathbb{N} is the type of natural numbers. Dependent typed programming would be very verbose if we always had to explicitly mention all the parameters. In cases when some parameters are inferable from the context, the programmer may leave them out, as in *id* _3.

For brevity, Agda supports implicit parameters. In the definition below:

```
id : {A : Set} → A → A
id a = a,
```

the parameter $\{A : \text{Set}\}$ in curly brackets is implicit and need not be mentioned when *id* is called, e.g., *id* 3. Agda tries to infer implicit parameters whenever possible. In case the inference fails, they can be explicitly provided in curly brackets: *id* $\{\mathbb{N}\}$ 3.

Named parameters in the type signature can be collected in a *telescope*. For example, $\{x : A\} \rightarrow \{y : A\} \rightarrow (z : B) \rightarrow \{w : C\} \rightarrow D$ can be abbreviated to $\{x\ y : A\}(z : B)\{w : C\} \rightarrow D$.

As an example of a datatype definition, cons-lists can be defined by:

```
data [] (A : Set) : Set where
  []      : [A]
  _::_    : A → [A] → [A].
```

In Agda’s notation for dist-fix definitions, an underline denotes a location for a parameter. The type constructor $[_]$ takes a type and yields a type. The parameter $(A : \text{Set})$, written on the left-hand side of the colon, scopes over the entire definition and is an implicit parameter of the constructors $_{::}$ and $[_]$.

2.1 First-Order Logic

In the Curry-Howard isomorphism, types are propositions and terms their proofs. Being able to construct a term of a particular type is to provide a proof of that proposition. Fig. 2 shows an encoding of first-order intuitionistic logic in Agda. Falsity is represented by \perp , a type with no constructors and therefore no inhabitants. Truth, on the other hand, can be represented by the type \top , having one unique term — a record with no fields. Disjunction is represented by disjoint sum, while conjunction is denoted by product as usual: a proof of $P \uplus Q$ can be deducted either from a proof of P or a proof of Q , while a proof of $P \times Q$ consists of proofs of both.

An implication $P \rightarrow Q$ is represented as a function taking a proof of P to a proof of Q . We do not introduce new notation for it. The quantifier \forall is encoded as a dependent function which, given any $x : A$, must produce a proof of $P\ x$. Agda provides a short hand *forall* $x \rightarrow P\ x$ in place of $(x : A) \rightarrow P\ x$ when A can be inferred. To prove the proposition $\exists A\ P$, where P is a predicate on terms of type A , one has to provide a witness $w : A$ and a proof of $P\ w$. Given a term of type $\exists A\ P$, the two functions *witness* and *proof* extract the witness and the proof, respectively.

<pre> data \perp : <i>Set</i> where record \top : <i>Set</i> where data $_ \uplus _$ (<i>P Q</i> : <i>Set</i>) : <i>Set</i> where <i>inj</i>₁ : <i>P</i> → <i>P</i> \uplus <i>Q</i> <i>inj</i>₂ : <i>Q</i> → <i>P</i> \uplus <i>Q</i> data $_ \times _$ (<i>P Q</i> : <i>Set</i>) : <i>Set</i> where $_ \times _$: <i>P</i> → <i>Q</i> → <i>P</i> \times <i>Q</i> </pre>	<pre> data \exists (<i>A</i> : <i>Set</i>) (<i>P</i> : <i>A</i> → <i>Set</i>) : <i>Set</i> where <i>exists</i> : (<i>w</i> : <i>A</i>) → <i>P w</i> → \exists <i>A P</i> <i>witness</i> : {<i>A</i> : <i>Set</i>} {<i>P</i> : <i>A</i> → <i>Set</i>} → \exists <i>A P</i> → <i>A</i> <i>witness</i> (<i>exists w p</i>) = <i>w</i> <i>proof</i> : {<i>A</i> : <i>Set</i>} {<i>P</i> : <i>A</i> → <i>Set</i>} → (<i>x</i> : \exists <i>A P</i>) → <i>P</i> (<i>witness x</i>) <i>proof</i> (<i>exists w p</i>) = <i>p</i> </pre>
--	---

Fig. 2. An encoding of first-order intuitionistic logic in Agda.

2.2 Identity Type

A term of type $x \equiv y$ is a proof that the values x and y are equal. The datatype \equiv is defined by:

```

data  $\equiv$  {A : Set} (x : A) : A → Set where
   $\equiv$ -refl : x  $\equiv$  x.

```

Agda has relaxed lexical rules allowing Unicode characters in identifiers. Therefore, \equiv -*refl* (without space) is a valid name. Since the only constructor \equiv -*refl* is of type $x \equiv x$, being able to type-check a term with type $x \equiv y$ means that the type checker is able to deduce that x and y are indeed equal⁴.

For the rest of the paper, we will exploit Unicode characters to give telling names to constructors, arguments, and lemmas. For example, if a variable is a proof of $y \equiv z$, we may name it $y \equiv z$ (without space).

The type \equiv is reflexive by definition. It is also symmetric and transitive, meaning that given a term of type $x \equiv y$, one can construct a term of type $y \equiv x$, and that given $x \equiv y$ and $y \equiv z$, one can construct $x \equiv z$:

```

 $\equiv$ -sym : {A : Set} {x y : A} → x  $\equiv$  y → y  $\equiv$  x
 $\equiv$ -sym  $\equiv$ -refl =  $\equiv$ -refl,

 $\equiv$ -trans : {A : Set} {x y z : A} → x  $\equiv$  y → y  $\equiv$  z → x  $\equiv$  z
 $\equiv$ -trans  $\equiv$ -refl y  $\equiv$  z = y  $\equiv$  z.

```

The type of the first explicit parameter in the type signature of \equiv -*sym* is $x \equiv y$, while the constructor \equiv -*refl* in the pattern has type $x \equiv x$. When the type checker tries to unify them, x is unified with y . Therefore, when we need to return a term of type $y \equiv x$ on the right-hand side, we can simply return \equiv -*refl*. The situation with \equiv -*trans* is similar. Firstly, x is unified with y , therefore the parameter $y \equiv z$, having type $y \equiv z$, can also be seen as having type $x \equiv z$ and

⁴ Agda assume uniqueness of identity proofs (but not proof irrelevance).

```

infixr 2  $\sim\langle\_ \rangle\_$ 
infix 2  $\sim\Box$ 

 $\sim\langle\_ \rangle\_ : \{A : Set\}(x : A)\{y z : A\} \rightarrow x \sim y \rightarrow y \sim z \rightarrow x \sim z$ 
 $x \sim\langle x \sim y \rangle y \sim z = \sim\text{-trans } x \sim y \ y \sim z$ 

 $\sim\Box : \{A : Set\}(x : A) \rightarrow x \sim x$ 
 $x \sim\Box = \sim\text{-refl}$ 

```

Fig. 3. Combinators for preorder reasoning.

be returned. In general, pattern matching and inductive families (such as \equiv) is a very powerful combination.

The interactive feature of Agda⁵ is helpful for constructing the proof terms. One may, for example, leave out the right-hand side as an interaction point. Agda would prompt the programmer with the expected type of the term to fill in, which also corresponds to the remaining proof obligations. The list of variables in the current context and their types after unification are also available to the programmer.

The lemma $\equiv\text{-subst}$ states that Leibniz equality holds: if $x \equiv y$, they are interchangeable in all contexts. Given a context f and a proof that $x \equiv y$, the congruence lemma $\equiv\text{-cong}$ returns a proof that $f x \equiv f y$.

```

 $\equiv\text{-subst} : \{A : Set\}(P : A \rightarrow Set)\{x y : A\} \rightarrow x \equiv y \rightarrow P x \rightarrow P y$ 
 $\equiv\text{-subst } P \equiv\text{-refl } P x = P x,$ 

 $\equiv\text{-cong} : \{A B : Set\}(f : A \rightarrow B)\{x y : A\} \rightarrow x \equiv y \rightarrow f x \equiv f y$ 
 $\equiv\text{-cong } f \equiv\text{-refl} = \equiv\text{-refl}.$ 

```

2.3 Preorder Reasoning

To prove a proposition $e_1 \equiv e_2$ is to construct a term having that type. One can do so by the operators defined in the previous section. It can be very tedious, however, when the expressions involved get complicated. Luckily, for any binary relation \sim that is reflexive and transitive (that is, for which one can construct terms $\sim\text{-refl}$ and $\sim\text{-trans}$ having the types as described in the previous section), we can induce a set of combinators, shown in Fig. 3, which allows one to construct a term of type $e_1 \sim e_n$ in algebraic style. These combinators are implemented in Agda by Norell [17] and improved by Danielsson in the Standard Library of Agda [21]. Augustsson [3] has proposed a similar syntax for equality reasoning, with automatic inference of congruences.

⁵ Agda has an Emacs mode and a command line interpreter interface.

$$\begin{aligned}
& \text{foldr-fusion} : \{A\ B\ C : \text{Set}\} \rightarrow (h : B \rightarrow C) \rightarrow \{f : A \rightarrow B \rightarrow B\} \rightarrow \\
& \quad \{g : A \rightarrow C \rightarrow C\} \rightarrow \{z : B\} \rightarrow (\text{push} : \text{forall } a \rightarrow h \cdot f\ a \dot{=} g\ a \cdot h) \rightarrow \\
& \quad h \cdot \text{foldr } f\ z \dot{=} \text{foldr } g\ (h\ z) \\
& \text{foldr-fusion } h\ \{f\}\ \{g\}\ \{z\} - [] = \equiv\text{-refl} \\
& \text{foldr-fusion } h\ \{f\}\ \{g\}\ \{z\}\ \text{push } (a :: as) = \\
& \quad \text{let } ih = \text{foldr-fusion } h\ \text{push } as \text{ in} \\
& \quad h\ (\text{foldr } f\ z\ (a :: as)) \\
& \equiv \langle \equiv\text{-refl} \rangle \\
& \quad h\ (f\ a\ (\text{foldr } f\ z\ as)) \\
& \equiv \langle \text{push } a\ (\text{foldr } f\ z\ as) \rangle \\
& \quad g\ a\ (h\ (\text{foldr } f\ z\ as)) \\
& \equiv \langle \equiv\text{-cong } (g\ a)\ ih \rangle \\
& \quad g\ a\ (\text{foldr } g\ (h\ z)\ as) \\
& \equiv \langle \equiv\text{-refl} \rangle \\
& \quad \text{foldr } g\ (h\ z)\ (a :: as) \\
& \equiv \square
\end{aligned}$$

Fig. 4. Proving the fusion theorem for *foldr*.

To understand the definitions, notice that $\sim\langle_ \rangle\text{-}$, a dist-fix function taking three explicit parameters, associates to the right. Therefore, the algebraic proof:

$$\begin{aligned}
& e_1 \\
& \sim \langle \text{reason}_1 \rangle \\
& \quad \vdots \\
& e_{n-1} \\
& \sim \langle \text{reason}_{n-1} \rangle \\
& e_n \\
& \sim \square
\end{aligned}$$

should be bracketed as $e_1 \sim \langle \text{reason}_1 \rangle \dots (e_{n-1} \sim \langle \text{reason}_{n-1} \rangle (e_n \sim \square))$. Each occurrence of $\sim\langle_ \rangle\text{-}$ takes three arguments: e_i on the left, reason_i (a proof that $e_i \sim e_{i+1}$) in the angle brackets, and a proof of $e_{i+1} \sim e_n$ on the right-hand side, and produces a proof of $e_i \sim e_n$ using $\sim\text{-trans}$. As the base case, $\sim\square$ takes the value e_n and returns a term of type $e_n \sim e_n$.

2.4 Functional Derivation

The ingredients we have prepared so far already allow us to perform some functional program derivation. For brevity, however, we introduce an equivalence relation on functions:

$$\begin{aligned}
& \dot{=}_- : \{A\ B : \text{Set}\} \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow B) \rightarrow \text{Set} \\
& f \dot{=} g = \text{forall } a \rightarrow f\ a \equiv g\ a.
\end{aligned}$$

Since $\dot{=}_-$ can be shown to be reflexive and transitive, it also induces a set of pre-order reasoning operators. Fig. 4 shows a proof of the foldr fusion theorem. The

$$\begin{aligned}
& \text{scanr-der} : \{A B : \text{Set}\} \rightarrow (f : A \rightarrow B \rightarrow B) \rightarrow (e : B) \rightarrow \\
& \quad \exists ([A] \rightarrow \text{List}^+ B) (\backslash \text{prog} \rightarrow \text{map}^+ (\text{foldr } f \ e) \cdot \text{tails} \doteq \text{prog}) \\
& \text{scanr-der } f \ e = \text{exists } _ \left(\text{map}^+ (\text{foldr } f \ e) \cdot \text{tails} \right. \\
& \quad \quad \quad \doteq \langle \text{foldr-fusion } (\text{map}^+ (\text{foldr } f \ e)) (\text{push-map-til } f) \ \rangle \\
& \quad \quad \quad \text{foldr } (sc \ f) [e]^+ \\
& \quad \quad \quad \left. \doteq \square \right) \\
& \text{where } sc : \{A B : \text{Set}\} \rightarrow (A \rightarrow B \rightarrow B) \rightarrow A \rightarrow \text{List}^+ B \rightarrow \text{List}^+ B \\
& \quad \quad \quad sc \ f \ a [b]^+ = f \ a \ b ::^+ [b]^+ \\
& \quad \quad \quad sc \ f \ a (b ::^+ bs) = f \ a \ b ::^+ b ::^+ bs \\
& \text{push-map-til} : \{A B : \text{Set}\} \rightarrow (f : A \rightarrow B \rightarrow B) \rightarrow \{e : B\} \rightarrow (a : A) \rightarrow \\
& \quad \quad \quad \text{map}^+ (\text{foldr } f \ e) \cdot \text{til } a \doteq sc \ f \ a \cdot \text{map}^+ (\text{foldr } f \ e) \\
& \text{push-map-til } f \ a [xs]^+ = \equiv\text{-refl} \\
& \text{push-map-til } f \ a (xs ::^+ xss) = \equiv\text{-refl}
\end{aligned}$$

Fig. 5. Derivation of *scanr*. The constructors $::^+$ and $[_] ^+$ build non-empty lists, while $\text{tails} = \text{foldr } \text{til } []^+$, where $\text{til } a [xs]^+ = (a :: xs) ::^+ [xs]^+$; $\text{til } a (xs ::^+ xss) = (a :: xs) ::^+ xs ::^+ xss$.

steps using $\equiv\text{-refl}$ are simple equivalences which Agda can prove by expanding the definitions. The inductive hypothesis *ih* is established by a recursive call. Agda ensures that proofs by induction are well-founded. Fig. 5 derives *scanr* from its specification $\text{map}^+ (\text{foldr } f \ e) \cdot \text{tails}$, where map^+ is the map function defined for List^+ , the type of non-empty lists. The *foldr-fusion* theorem is used to transform the specification to a fold. The derived program can be extracted by $\text{scanr} = \text{witness } \text{scanr-der}$, while $\text{scanr-pf} = \text{proof } \text{scanr-der}$ is a proof that can be used elsewhere. Notice that the first argument to *exists* is left implicit. Agda is able to infer the witness because it is syntactically presented in the derivation.

We have reproduced a complete derivation for the maximum segment sum problem. The derivation proceeds in the standard manner [6], transforming the specification to $\text{max} \cdot \text{map } (\text{foldr } _ \otimes _ 0) \cdot \text{tails}$ for some $_ \otimes _$, and exploiting *scanr-pf* to convert it to a *scanr*. The main derivation is about 220 lines long, plus 400 lines of library code proving properties about lists and 100 lines for properties about integers. The code is available online [16].

The interactive interface of Agda proved to be very useful. One could progress the derivation line by line, leaving out the unfinished part as an interaction point. One may also type in the desired next step but leave the “reason” part blank, and let Agda derive the type of the lemma needed.

3 Relational Derivation

During the 90’s there was a trend in the program derivation community to move from functions to relations. For completeness, we give a quick introduction to relations in this section. The reader is referred to Backhouse et al. [4] and Backhouse and Hoogendijk [5] for a more rigorous treatment. Bird and de Moor [8]

present program derivation from a more abstract, category-theoretical point of view, with many illustrative examples of program derivation.

A relation R to B from A , denoted by $R : B \leftarrow A$, is usually understood as a subset of the set of pairs $A \times B$.⁶ A function f is seen as a special case where $(a, b) \in f$ and $(a, b') \in f$ implies $b = b'$. The use of relations allows non-determinism in the specification. Derivation proceeds by inclusion as well as equality: in each step, the specification may be refined to a more deterministic subset, often all the way until we reach a function.

The composition of two relations $R : C \leftarrow B$ and $S : B \leftarrow A$ is defined by: $R \circ S = \{(a, c) \mid \exists b : (a, b) \in S \wedge (b, c) \in R\}$. Given a relation $R : B \leftarrow A$, its power transpose ΛR is a function from A to $\mathbf{P}B$ (subsets of B): $\Lambda R a = \{b \mid (a, b) \in R\}$, while the relation $\in : A \leftarrow \mathbf{P}A$ maps a set to one of its arbitrary members.

The fold remains an important construct in the relational setting. While foldr takes a step function of type $A \rightarrow B \rightarrow B$ and a base case of type B , its relational counterpart, which we denote by foldR , takes an uncurried relation $R : B \leftarrow (A \times B)$, while the base cases, being non-deterministic, are recorded in a set $s : \mathbf{P}B$.⁷ The relational fold can be defined in terms of functional fold:

$$\begin{aligned} \text{foldR } R \ s &: B \leftarrow [A] \\ \text{foldR } R \ s &= \in \circ \text{foldr } \Lambda(R \circ (\text{id} \times \in)) \ s. \end{aligned}$$

We will see in the next few sections how these concepts can be modelled in Agda.

3.1 Modelling Relations

A possibly infinite subset (of A) could be represented by its membership function of type $\mathbf{P}A = A \rightarrow \text{Bool}$. With dependent types, we can also represent the membership judgement at type level:

$$\begin{aligned} \mathbf{P} &: \text{Set} \rightarrow \text{Set1} \\ \mathbf{P}A &= A \rightarrow \text{Set}. \end{aligned}$$

A set $s : \mathbf{P}A$ is a function mapping $a : A$ to a type, which encodes a logic formula determining its membership. Agda maintains a hierarchy of universes, where Set denotes the universe of types, Set1 denotes the universe of Set and all types declared as being in Set1 , etc. Since $s : \mathbf{P}A$ is a function yielding a Set , $\mathbf{P}A$ is in the universe Set1 . The function *singleton* creates singleton sets:

$$\begin{aligned} \text{singleton} &: \{A : \text{Set}\} \rightarrow A \rightarrow \mathbf{P}A \\ \text{singleton } a &= \lambda a' \rightarrow a \equiv a'. \end{aligned}$$

Set union and inclusion, for example, are naturally encoded by disjunction and implication:

⁶ Notations used in the beginning of this section, for example \times , \in , and set comprehension, refer to their usual set-theoretical definitions. We will talk about how they can be represented in Agda in the next few subsections.

⁷ Isomorphically, the base case can be represented by a relation $B \leftarrow \top$.

$$\begin{array}{ll} _ \sqcup _ : \{A : Set\} \rightarrow PA \rightarrow PA \rightarrow PA & _ \sqsubseteq _ : \{A : Set\} \rightarrow PA \rightarrow PA \rightarrow Set \\ r \sqcup s = \lambda a \rightarrow r a \sqcup s a, & r \sqsubseteq s = \text{forall } a \rightarrow r a \rightarrow s a. \end{array}$$

A term of type $r \sqsubseteq s$ is a function which, given an a and a proof that a is in r , produces a proof that a is in s .

A relation $B \leftarrow A$, seen as a set of pairs, could be represented as $P(A \times B) = (A \times B) \rightarrow Set$. However, we find the following “curried” representation more convenient:

$$\begin{array}{l} _ \leftarrow _ : Set \rightarrow Set \rightarrow Set1 \\ B \leftarrow A = A \rightarrow B \rightarrow Set. \end{array}$$

One of the advantages is that relations and set-valued functions are unified. The A operator, for example, is simply the identity function at the term-level:

$$\begin{array}{l} A : \{A B : Set\} \rightarrow (B \leftarrow A) \rightarrow (A \rightarrow PB) \\ AR = R. \end{array}$$

A function can be converted to a relation:

$$\begin{array}{l} \text{fun} : \{A B : Set\} \rightarrow (A \rightarrow B) \rightarrow (B \leftarrow A) \\ \text{fun } f a b = f a \equiv b. \end{array}$$

The identity relation, for example, is denoted $id_R : \{A : Set\} \rightarrow (A \leftarrow A)$ and defined by $id_R = \text{fun } id$.

Relational composition could be defined by $R \circ S = \exists B (\lambda b \rightarrow (S a b \times R b c))$. For reasons that will be clear in the next section, we split the definition into two parts, shown in Fig. 6. The operator $_ \circ _$ applies a relation $R : B \leftarrow A$ to a set PA , yielding another set PB . Composition $_ \circ _$ is then defined using $_ \circ _$.

Complication arises when we try to represent \in . Recall that \in maps PA to A . However, the second argument to $_ \leftarrow _$ must be in Set , while PA is in $Set1$! At present, we have no choice but to declare another type of arrows that accepts $Set1$ -sorted inputs:

$$\begin{array}{l} _ \leftarrow_1 _ : Set \rightarrow Set1 \rightarrow Set1 \\ B \leftarrow_1 PA = PA \rightarrow B \rightarrow Set. \end{array}$$

It means we need several alternatives of relational composition that differ only in their types. Fig. 6 shows $_ \circ_1 _$ and $_ \circ _$ for example. Such inconvenience may be resolved if Agda introduces *universe polymorphism*, a feature on the wish-list at the time of writing. Also summarised in Fig. 6 are $_ \smile$ for relational converse, and $_ \times_1 _$, a higher-kinded variation of the product functor.

3.2 Inclusion and Monotonicity

A relation S can be refined to R if every possible outcome of R is a legitimate outcome of S . We represent the refinement relation by:

$$\begin{array}{l} _ \sqsubseteq _ : \{A B : Set\} \rightarrow (B \leftarrow A) \rightarrow (B \leftarrow A) \rightarrow Set \\ R \sqsubseteq S = \text{forall } a \rightarrow R a \sqsubseteq S a, \end{array}$$

$$\begin{aligned}
\check{_} : \{A B : Set\} &\rightarrow (A \leftarrow B) \rightarrow (B \leftarrow A) & \in : \{A : Set\} &\rightarrow (A \leftarrow_1 \mathsf{P}A) \\
R^\check{_} = \backslash a b &\rightarrow R b a & \in &= \backslash pa a \rightarrow pa a \\
\\
_\bullet : \{A B : Set\} &\rightarrow (B \leftarrow A) \rightarrow \mathsf{P}A \rightarrow \mathsf{P}B \\
R_\bullet s = \backslash b &\rightarrow \exists A (\backslash a \rightarrow (s a \times R a b)) \\
\\
\circ : \{A B C : Set\} &\rightarrow (C \leftarrow B) \rightarrow (B \leftarrow A) \rightarrow (C \leftarrow A) \\
(R \circ S) a &= R_\bullet (S a) \\
\\
_\circ_1_ : \{A : Set1\} \{B C : Set\} &\rightarrow (C \leftarrow B) \rightarrow (B \leftarrow_1 A) \rightarrow (C \leftarrow A) \\
(R \circ_1 S) a &= R_\bullet (S a) \\
\\
_\circ_1_ : \{A B C : Set\} &\rightarrow (C \leftarrow_1 \mathsf{P}B) \rightarrow (B \leftarrow A) \rightarrow (C \leftarrow A) \\
(R \circ_1 S) a &= R (S a) \\
\\
_\times_1_ : \{A B : Set\} \{PC : Set1\} \{D : Set\} &\rightarrow \\
&(B \leftarrow A) \rightarrow (D \leftarrow_1 PC) \rightarrow ((B \times D) \leftarrow_1 (A \times_1 PC)) \\
(R \times_1 S) (a, pc) (b, d) &= R a b \times S pc d
\end{aligned}$$

Fig. 6. Some operators on \leftarrow and \leftarrow_1 relations, including composition, membership and product. In this paper, \times_1 is overloaded for the type of pairs whose right component is in *Set1* (\circ_1 being the data constructor), and its functor action on relations (defined in this figure).

which expands to *forall* $a \rightarrow$ *forall* $b \rightarrow R a b \rightarrow S a b$. Conversely, $R \sqsubseteq S = S \sqsubseteq R$. Both \sqsubseteq and \sqsupseteq can be shown to be reflexive and transitive. Therefore, we can use them for preorder reasoning.

In hand-written derivation, the monotonicity of \circ (that is, $S \sqsubseteq T$ implies $R \circ S \sqsubseteq R \circ T$) is often used without being explicitly stated. In our Agda encoding where there are many versions of composition, it appears that we need one monotonicity lemma for each of them. Luckily, since those alternatives of composition are all defined in terms of $_\bullet$, it is enough to model monotonicity for $_\bullet$ only:

$$\begin{aligned}
_\bullet\text{-monotonic} : \{A B : Set\} &\rightarrow (R : B \leftarrow A) \rightarrow \{s t : \mathsf{P}A\} \rightarrow \\
&s \sqsubseteq t \rightarrow R_\bullet s \sqsubseteq R_\bullet t \\
_\bullet\text{-monotonic } R s \sqsubseteq t &b \text{ (exists } a_1 (a_1 \in s, b R a_1)) = \\
&\text{exists } a_1 (s \sqsubseteq t a_1 a_1 \in s, b R a_1).
\end{aligned}$$

To refine $R \circ S \circ T$ to $R \circ U \circ T$ given $U \sqsubseteq S$, for example, we may use $(\backslash x \rightarrow _\bullet\text{-monotonic } R (U \sqsubseteq S (T_\bullet x)))$ as the reason. It is instructive to study the definition of $_\bullet\text{-monotonic}$. After taking R and $s \sqsubseteq t$ (a proof of $s \sqsubseteq t$), the function $_\bullet\text{-monotonic}$ shall return a proof of $R_\bullet s \sqsubseteq R_\bullet t$. The proof, given a value b and a proof that some a_1 in s is mapped to b through R , shall produce a proof that there exists some value in t that is also mapped to b . The obvious

choice of such a value is a_1 . Notice how we apply $s \subseteq t$ to a_1 and $a_1 \in s$ to produce a proof that a_1 is also in t .

Another lemma often used without being said is that we can introduce id_R anywhere we need. It can be proved using $\equiv\text{-subst}$:

$$\begin{aligned} id\text{-intro} : \{A\ B : Set\} \{R : B \leftarrow A\} &\rightarrow R \sqsupseteq R \circ id_R \\ id\text{-intro} \{-\} \{-\} \{R\} \ a \ b \ (exists\ a' \ (a \equiv a', bRa')) &= \\ \equiv\text{-subst} \ (\backslash a \rightarrow R \ a \ b) \ (\equiv\text{-sym} \ a \equiv a') \ bRa'. \end{aligned}$$

3.3 Relational Fold

Having defined all the necessary components, we can now define relational fold in terms of functional fold:

$$\begin{aligned} foldR : \{A\ B : Set\} &\rightarrow (B \leftarrow (A \times B)) \rightarrow PB \rightarrow (B \leftarrow [A]) \\ foldR \ R \ s &= foldr \ (R \circ_1 \ (id_R \times_1 \in)) \ s. \end{aligned}$$

On the top of the list of properties that we wish to have proved is, of course, fold fusion for relational folds:

$$\begin{aligned} foldR\text{-fusion} : \{A\ B\ C : Set\} &\rightarrow (R : C \leftarrow B) \rightarrow \{S : B \leftarrow (A \times B)\} \rightarrow \\ &\{T : C \leftarrow (A \times C)\} \{u : PB\} \{v : PC\} \rightarrow \\ R \circ S &\sqsupseteq T \circ (id_R \times R) \rightarrow R \cdot u \sqsupseteq v \rightarrow \\ R \circ foldR \ S \ u &\sqsupseteq foldR \ T \ v. \end{aligned}$$

The proof proceeds by converting both sides to functional folds. It is omitted here for brevity but is available online [16]. To use fold fusion, however, there has to be a fold to start with. Luckily, this is hardly a problem, given the following lemma showing that id_R , when instantiated to lists, is a fold:

$$id_R \sqsupseteq foldR : \{A : Set\} \rightarrow id_R \ \{[A]\} \sqsupseteq foldR \ cons \ nil,$$

where $cons = fun \ (uncurry \ ::) \cdot$ and $nil = singleton \ []$. Let us try to construct its proof term. The inclusion $id_R \ \{[A]\} \sqsupseteq foldR \ cons \ nil$ expands to:

$$forall \ xs \ ys \rightarrow foldR \ cons \ nil \ xs \ ys \rightarrow xs \equiv ys.$$

The proof term of $id_R \sqsupseteq foldR$ should be a function which takes xs , ys , and a proof that $foldR \ cons \ nil$ maps xs to ys , and returns a proof of $xs \equiv ys$. When xs is $[]$, $foldR \ cons \ nil \ [] \ ys$ simplifies to $[] \equiv ys$, and we can simply return the proof:

$$id_R \sqsupseteq foldR \ [] \ ys \ [] \equiv ys = [] \equiv ys.$$

Consider the case $a :: xs$. The proposition $foldR \ cons \ nil \ (a :: xs) \ ys$ expands to $\exists (V \times [V]) \ P$, where $P \ (a', as) = ((a \equiv a') \times (foldR \ cons \ nil \ xs \ as)) \times (cons \ (a, as) \ ys)$.

Given $a :: xs$, ys , and a proof of $\exists (V \times [V]) P$, we should construct a proof that $a :: xs \equiv ys$. We can do so by equational reasoning:

$$\begin{aligned}
& id_R \sqsubseteq foldR (a :: xs) \ ys \ (exists (a', as) ((a \equiv a', foldR xs as), a' :: as \equiv ys)) = \\
& \quad a :: xs \\
& \equiv \langle \quad a \equiv a' \ \langle :: \rangle \ (id_R \sqsubseteq foldR \ xs \ as \ foldR xs as) \quad \rangle \\
& \quad a' :: as \\
& \equiv \langle \quad a' :: as \equiv ys \quad \rangle \\
& \quad ys \\
& \equiv \square,
\end{aligned}$$

where $\langle :: \rangle$ is $\equiv\text{-cong}$ applied twice, substituting a for a' and xs for as .

4 Example: Deriving Insertion Sort

We are finally in a position to present our main example: a derivation of insertion sort, adopted from Bird [7].

4.1 Specifying Sort

We first specify what a sorted list is, assuming a datatype Val and a binary ordering $_ \leq _ : Val \rightarrow Val \rightarrow Set$ that form a decidable total order. To begin with, let $lbound$ be the set of all pairs (a, xs) such that a is a lower bound of xs :

$$\begin{aligned}
lbound & : P(Val \times [Val]) \\
lbound (a, []) & = \top \\
lbound (a, b :: xs) & = (a \leq b) \times lbound (a, xs).
\end{aligned}$$

A *coreflexive* is a sub-relation of id_R . The following operator $_?$ converts a set to a coreflexive, letting the input go through iff it is in the set:

$$\begin{aligned}
_? & : \{A : Set\} \rightarrow PA \rightarrow (A \leftarrow A) \\
(p?) \ a \ b & = (a \equiv b) \times p \ a.
\end{aligned}$$

The coreflexive $ordered?$, which lets a list go through iff it is sorted, can then be defined as a fold:

$$\begin{aligned}
ordered? & : [Val] \leftarrow [Val] \\
ordered? & = foldR (cons \circ lbound?) \ nil.
\end{aligned}$$

We postulate a datatype Bag , representing bags of values. Bags are formed by two constructors: $\wr : Bag$ and $::_b : Val \rightarrow Bag \rightarrow Bag$. For the derivation to work, we demand that the result of $::_b$ be distinguishable from the empty bag, and that $::_b$ be commutative:⁸

$$\begin{aligned}
& ::_b\text{-nonempty} : forall \{a \ w\} \rightarrow (\wr \equiv a ::_b \ w) \rightarrow \perp \\
& ::_b\text{-commute} : (a \ b : Val) \rightarrow (w : Bag) \rightarrow a ::_b (b ::_b \ w) \equiv b ::_b (a ::_b \ w).
\end{aligned}$$

⁸ We can put more constraints on bags, such as that $::_b$ discards no elements. But the two properties are enough to guarantee that $isort$ is included in $ordered? \circ permute$.

The function *bagify*, defined below, converts a list to a bag by a fold:

$$\begin{aligned} \text{bagify} & : [Val] \rightarrow Bag \\ \text{bagify} & = \text{foldr } _::\text{b-} \rfloor \rfloor. \end{aligned}$$

To map a list to one of its arbitrary permutations, we simply convert it to a bag, and convert the bag back to a list! To sort a list is to find one of its permutations that is sorted:

$$\begin{aligned} \text{permute} & : [Val] \leftarrow [Val] \\ \text{permute} & = (\text{fun bagify})^\circ \circ \text{fun bagify}, \end{aligned}$$

$$\begin{aligned} \text{sort} & : [Val] \leftarrow [Val] \\ \text{sort} & = \text{ordered?} \circ \text{permute}. \end{aligned}$$

Thus completes the specification, from which we shall derive an algorithm that actually sorts a list.

4.2 The Derivation

The derivation begins with observing that *permute* can be turned into a fold. We first introduce an *id_R* by *id-intro*, followed by the lemma *id_R ⊒ foldR*, and fold fusion:

$$\begin{aligned} \text{perm-der} & : \exists_1 ([Val] \leftarrow [Val]) (\backslash \text{perm} \rightarrow \text{permute} \sqsupseteq \text{perm}) \\ \text{perm-der} & = \text{exists}_1 - (\quad \text{permute} \quad) \\ & \quad \sqsupseteq \langle \quad \text{id-intro} \quad \rangle \\ & \quad \text{permute} \circ \text{id}_R \\ & \quad \sqsupseteq \langle \quad (\backslash xs \rightarrow _ \cdot \text{monotonic permute} (\text{id}_R \sqsupseteq \text{foldR } xs)) \quad \rangle \\ & \quad \text{permute} \circ \text{foldR cons nil} \\ & \quad \sqsupseteq \langle \quad \text{foldR-fusion permute perm-step perm-base} \quad \rangle \\ & \quad \text{foldR combine nil} \\ & \quad \sqsupseteq \square, \end{aligned}$$

where \exists_1 is a *Set1* variant of \exists , with extraction functions *witness₁* and *proof₁*. The relation *combine* can be defined as follows:

$$\begin{aligned} \text{combine} & : [Val] \leftarrow (Val \times [Val]) \\ \text{combine} (a, xs) & = \text{cons} (a, xs) \cup \text{combine}' (a, xs), \\ \text{combine}' & : [Val] \leftarrow (Val \times [Val]) \\ \text{combine}' (a, []) & = \backslash ys \rightarrow \perp \\ \text{combine}' (a, b :: xs) & = (\backslash zs \rightarrow \text{cons} (b, zs)) \cdot \text{combine} (a, xs). \end{aligned}$$

Given (a, xs) , it inserts *a* into an arbitrary position of *xs*. For the *foldR-fusion* to work, we have to provide two proofs:

$$\begin{aligned} \text{perm-step} & : \text{permute} \circ \text{cons} \sqsupseteq \text{combine} \circ (\text{id}_R \times \text{permute}) \\ \text{perm-base} & : \text{permute} \cdot \text{nil} \sqsupseteq \text{nil}. \end{aligned}$$

But the real work is done in proving that shuffling the input list does not change the result of *bagify*:

$$\begin{aligned} \text{bagify-homo} : (a : \text{Val}) \rightarrow (xs\ ys : [\text{Val}]) \rightarrow \\ \text{combine } (a, xs)\ ys \rightarrow \text{bagify } (a :: xs) \equiv \text{bagify } ys. \end{aligned}$$

It is when proving this lemma that we need ::_b -*commute*.

After the reasoning above, we have at our hands:

$$\begin{aligned} \text{perm} &: [\text{Val}] \leftarrow [\text{Val}] \\ \text{perm} &= \text{witness}_1 \text{perm-der}, \\ \text{permute-is-fold} &: \text{permute} \sqsupseteq \text{perm} \\ \text{permute-is-fold} &= \text{proof}_1 \text{perm-der}. \end{aligned}$$

Therefore, $\text{perm} = \text{foldR combine nil}$, while *permute-is-fold* is a proof that *perm* refines *permute*.

Now that *permute* can be refined to a fold, a natural step to try is to fuse *ordered?* into the fold. We derive:

$$\begin{aligned} \text{sort-der} &: \exists ([\text{Val}] \rightarrow [\text{Val}]) (\backslash f \rightarrow \text{ordered?} \circ \text{permute} \sqsupseteq \text{fun } f) \\ \text{sort-der} &= \text{exists_} \\ &\quad (\quad \text{ordered?} \circ \text{permute} \\ &\quad \sqsupseteq \langle \quad (\backslash xs \rightarrow \text{monotonic ordered? } (\text{permute-is-fold } xs)) \quad \rangle \\ &\quad \text{ordered?} \circ \text{perm} \\ &\quad \sqsupseteq \langle \quad \sqsupseteq\text{-refl} \quad \rangle \\ &\quad \text{ordered?} \circ \text{foldR combine nil} \\ &\quad \sqsupseteq \langle \quad \text{foldR-fusion ordered? ins-step ins-base} \quad \rangle \\ &\quad \text{foldR } (\text{fun } (\text{uncurry insert})) \text{ nil} \\ &\quad \sqsupseteq \langle \quad \text{foldR-to-foldr insert } [] \quad \rangle \\ &\quad \text{fun } (\text{foldr insert } []) \\ &\quad \sqsupseteq \square). \end{aligned}$$

The function *insert* follows the usual definition:

$$\begin{aligned} \text{insert} &: \text{Val} \rightarrow [\text{Val}] \rightarrow [\text{Val}] \\ \text{insert } a \ [] &= a :: [] \\ \text{insert } a \ (b :: xs) &\textbf{with } a \leq? b \\ \dots \quad | \text{ yes } a \leq b &= a :: b :: xs \\ \dots \quad | \text{ no } a \not\leq b &= b :: \text{insert } a \ xs, \end{aligned}$$

where $a \leq? b$ determines whether $a \leq b$, whose result is case-matched by the **with** notation. The fusion conditions are:

$$\begin{aligned} \text{ins-step} &: \text{ordered?} \circ \text{combine} \sqsupseteq \text{fun } (\text{uncurry insert}) \circ (\text{id}_R \times \text{ordered?}) \\ \text{ins-base} &: \text{ordered?} \bullet \text{nil} \sqsupseteq \text{nil}. \end{aligned}$$

Finally, *foldR-to-foldr* is a small lemma allowing us to convert a relational fold to a functional fold, provided that its arguments have been refined to a function and a singleton set already:

$$\begin{aligned} \text{foldR-to-foldr} &: \{A\ B : \text{Set}\} \rightarrow (f : A \rightarrow B \rightarrow B) \rightarrow (e : B) \rightarrow \\ &\quad \text{foldR } (\text{fun } (\text{uncurry } f)) (\text{singleton } e) \sqsupseteq \text{fun } (\text{foldr } f\ e). \end{aligned}$$

We have thus derived $isort = witness\ sort\text{-}der = foldr\ insert\ []$, while at the same time proved that it meets the specification $ordered? \circ permute$. The details of the proofs are available online [16]. The library code defining sets, relations, folds, and their properties, amounts to about 800 lines. The main derivation of $isort$ is not long. Proving the fusion condition $ins\text{-}step$ and its related properties turned out to take some hard work and eventually adds up to about 700 lines of code. The interactive mode was of great help — the proof would have been difficult to construct by hand.

5 Conclusion and Related Work

We have shown how to encode relational program derivation in a dependently typed language. Derivation is carried out in the host language, the correctness being guaranteed by the type system. It also demonstrates that dependent types are expressive enough to demand that a program satisfies an input/output relation. An interesting way to construct the corresponding proof term, which would be difficult to build otherwise, is derivation.

McKinna and Burstall’s paper on “Deliverables” [15] is an early example of machine checked program + proof construction (using Pollack’s LEGO). In their terminology $sort\text{-}der$ would be a deliverable — an element of a dependent Σ -type pairing up a function and a proof of correctness. In the Coq tradition Program Extraction has been used already from Paulin-Mohring’s early paper [18] to the impressive four-colour theorem development (including the development of a verified compiler). Our contribution is more modest — we aim at formally checked but still readable Algebra-of-Programming style derivations.

The concept of Inductive Families [12], especially the identity type $(=)$, is central to the Agda system and to our derivations. A recent development of relations in dependent type theory was carried out by Gonzalía [13, Ch. 5]. The advances in Agda’s notation and support for hidden arguments between that derivation and our work is striking.

There has been a trend in recent years to bridge the gap between dependent types and practical programming. Projects along this line include Cayenne [2], Coq [10], Dependent ML [23], Agda [17], Ω mega [19], Epigram [14], and the GADT extension [9] to Haskell. It is believed that dependent types have an important role in the next generation of programming languages [20].

Acknowledgements We are grateful to Nils Anders Danielsson for pointing out typos and giving valuable suggestions regarding the presentation.

References

1. T. Altenkirch, C. McBride, and J. McKinna. Why dependent types matter. Draft, 2005.
2. L. Augustsson. Cayenne – a language with dependent types. In *ICFP’98*, pages 239–250, 1998.

3. L. Augustsson. Equality proofs in Cayenne. Chalmers Univ. of Tech., 1999.
4. R. C. Backhouse et al. Relational catamorphisms. In *IFIP TC2/WG2.1 Working Conference on Constructing Programs*, pages 287–318. Elsevier, 1991.
5. R. C. Backhouse and P. F. Hoogendijk. Elements of a relational theory of datatypes. In B. Möller et al., editors, *Formal Program Development.*, number 755 in LNCS, pages 7–42. Springer-Verlag, January 1992.
6. R. S. Bird. Algebraic identities for program calculation. *Computer Journal*, 32(2):122–126, April 1989.
7. R. S. Bird. Functional algorithm design. *Science of Computer Programming*, 26:15–31, 1996.
8. R. S. Bird and O. de Moor. *Algebra of Programming*. International Series in Computer Science. Prentice Hall, 1997.
9. J. Cheney and R. Hinze. First-class phantom types. Technical Report TR2003-1901, Cornell University, 2003.
10. The Coq Development Team, LogiCal Project. *The Coq Proof Assistant Reference Manual*, 2006.
11. O. de Moor and G. Sittampalam. Higher-order matching for program transformation. *Theoretical Computer Science*, 269(1-2):135–162, 2001.
12. P. Dybjer. Inductive families. *Formal Aspects of Computing*, pages 440–465, 1994.
13. C. Gonzalía. *Relations in Dependent Type Theory*. PhD thesis, Chalmers Univ. of Tech., 2006.
14. C. McBride and J. McKinna. The view from the left. *Journal of Functional Programming*, 14(1):69–111, 2004.
15. J. McKinna and R. M. Burstall. Deliverables: A categorical approach to program development in type theory. In *MFCS’93: Int. Symp. on Mathematical Foundations of Computer Science*, pages 32–67. Springer-Verlag, 1993.
16. S.-C. Mu, H.-S. Ko, and P. Jansson. AoPA: Algebra of programming in Agda. <http://www.iis.sinica.edu.tw/~scm/2008/aopa/>.
17. U. Norell. *Towards a Practical Programming Language Based on Dependent Type Theory*. PhD thesis, Chalmers Univ. of Tech., 2007.
18. C. Paulin-Mohring. Extracting F_ω ’s programs from proofs in the Calculus of Constructions. In *POPL’89*, Austin, Jan. 1989. ACM.
19. T. Sheard. Programming in Ω mega. The 2nd Central European Functional Programming School, June 2007.
20. T. Sweeney. The next mainstream programming language: a game developer’s perspective. Invited Talk, POPL’06, January 2006.
21. The Agda Team. The Agda Wiki. Available at <http://www.cs.chalmers.se/~ulfn/Agda/>, 2007.
22. R. Verhoeven and R. C. Backhouse. Towards tool support for program verification and construction. In *World Congress on Formal Methods*, pages 1128–1146, 1999.
23. H. Xi. Dependent ML: an approach to practical programming with dependent types. *Journal of Functional Programming*, 17(2):215–286, March 2007.
24. T. Yokoyama, Z. Hu, and M. Takeichi. Yicho - a system for programming program calculations. In *The 3rd Asian Workshop on Programming Languages and Systems (APLAS 2002)*, pages 366–382, 2002.