

# Subtyping, Declaratively

## An Exercise in Mixed Induction and Coinduction

Nils Anders Danielsson and Thorsten Altenkirch

University of Nottingham

**Abstract.** It is natural to present subtyping for recursive types coinductively. However, Gapeyev, Levin and Pierce have noted that there is a problem with coinductive definitions of non-trivial transitive inference systems: they cannot be “declarative”—as opposed to “algorithmic” or syntax-directed—because coinductive inference systems with an explicit rule of transitivity are trivial.

We propose a solution to this problem. By using mixed induction and coinduction we define an inference system for subtyping which combines the advantages of coinduction with the convenience of an explicit rule of transitivity. The definition uses coinduction for the structural rules, and induction for the rule of transitivity. We also discuss under what conditions this technique can be used when defining other inference systems. The developments presented in the paper have been mechanised using Agda, a dependently typed programming language and proof assistant.

## 1 Introduction

Coinduction and corecursion are useful techniques for defining and reasoning about things which are potentially infinite, including streams and other (potentially) infinite data types (Coquand 1994; Giménez 1996; Turner 2004), process congruences (Milner 1990), congruences for functional programs (Gordon 1999), closures (Milner and Tofte 1991), semantics for divergence of programs (Cousot and Cousot 1992; Hughes and Moran 1995; Leroy and Grall 2009; Nakata and Uustalu 2009), and subtyping relations for recursive types (Brandt and Henglein 1998; Gapeyev et al. 2002).

However, the use of coinduction can lead to values which are “too infinite”. For instance, a non-trivial binary relation defined as a coinductive inference system cannot include the rule of transitivity, because a coinductive reading of transitivity would imply that every element is related to every other (to see this, build an infinite derivation consisting solely of uses of transitivity). As pointed out by Gapeyev et al. (2002) this is unfortunate, because without transitivity, conceptually unrelated rules may have to be merged or otherwise modified in order to ensure that transitivity can be proved as a derived property. Gapeyev et al. give the example of subtyping for records, where a dedicated rule of transitivity ensures that one can give separate rules for depth subtyping (which states that a record field type can be replaced by a subtype), width subtyping (which

states that new fields can be added to a record), and permutation of record fields.

We propose a solution to this problem. The problem stems from a *coinductive* reading of transitivity, and it can be solved by reading the rule of transitivity *inductively*, and only using coinduction where it is necessary. We illustrate this idea by using mixed induction and coinduction to define a subtyping relation for recursive types; such relations have been studied repeatedly in the past (Amadio and Cardelli 1993; Kozen et al. 1995; Brandt and Henglein 1998, and others). The rule which defines when a function type is a subtype of another is defined coinductively, following Brandt and Henglein (1998) and Gapeyev et al. (2002), while the rule of transitivity is defined inductively.

The technique of mixing induction and coinduction has been known for a long time (Park 1980; Barwise 1989; Raffalli 1994; Giménez 1996; Hensel and Jacobs 1997; Müller et al. 1999; Barthe et al. 2004; Levy 2006; Bradfield and Stirling 2007; Abel 2009; Hancock et al. 2009), but we feel that it deserves to be more well-known in the programming language community. We hope that this paper provides an example of the use of mixed induction and coinduction which is appealing to programming language researchers.

The main developments in the paper have been formalised using the dependently typed, total<sup>1</sup> functional programming language Agda (Norell 2007; Agda Team 2009), and the source code is at the time of writing available to download (Danielsson 2009).

The rest of the paper is structured as follows: Section 2 gives an introduction to induction and coinduction in the context of Agda. Section 3 defines a small language of recursive types, and Sect. 4 defines a subtyping relation for this language by viewing the types as potentially infinite trees. Section 5 defines an equivalent, declarative subtyping relation using mixed induction and coinduction, and Sect. 6 compares this definition to another equivalent definition, given by Brandt and Henglein (1998). Finally Sect. 7 discusses a potential pitfall associated with the technique we propose, and Sect. 8 concludes.

## 2 Induction and Coinduction

This section gives a brief introduction to induction and coinduction, with an emphasis on how these concepts are realised in Agda. For more formal accounts of induction and coinduction see, for instance, the theses of Hagino (1987) and Mendler (1988). For a more detailed discussion of mixed induction and coinduction in Agda, see Danielsson and Altenkirch (2009).<sup>2</sup>

---

<sup>1</sup> Agda is an experimental system. The meta-theory has not been formalised, and the type checker has not been proved bug-free, so take phrases such as “total” with a grain of salt.

<sup>2</sup> This unpublished draft contains large parts of the present paper.

## 2.1 Induction

Let us start with a simple inductive definition. In Agda the type of *finite* lists can be defined as follows:

```
data List (A : Set) : Set where
  []      : List A
  _::_   : A → List A → List A
```

This states that *List A* is a type (or *Set*) with two constructors, `[]` of type *List A* and `_::_` of type  $A \rightarrow \text{List } A \rightarrow \text{List } A$ . The constructor `_::_` is an infix operator; the underscores mark the argument positions. The type *List A* is isomorphic to the least fixpoint  $\mu X. 1 + A \times X$  in the category of types and total functions.<sup>3</sup>

Agda has a termination checker which ensures that all code is terminating (or productive, see below). The termination checker allows lists to be destructured using structural recursion:

```
map : {A B : Set} → (A → B) → List A → List B
map f []          = []
map f (x :: xs) = f x :: map f xs
```

The use of braces in  $\{A B : \text{Set}\} \rightarrow \dots$  means that the two type arguments *A* and *B* are *implicit*; they do not need to be given explicitly if Agda can infer them. (Note that in this context *A B* is not an application, it is a sequence of variables.)

## 2.2 Coinduction

If we want to have infinite lists, or streams, we can use the following coinductive definition instead (note that constructors, such as `_::_`, can be overloaded in Agda):

```
data Stream (A : Set) : Set where
  _::∞ : A → ∞ (Stream A) → Stream A
```

The type *Stream A* is isomorphic to the greatest fixpoint  $\nu X. A \times X$ . The type function  $\infty : \text{Set} \rightarrow \text{Set}$  marks its argument as being coinductive. It is analogous to the suspension type constructors which are sometimes used to implement non-strictness in strict languages (Wadler et al. 1998). The type comes with a force function and a delay constructor:

```
!_ : {A : Set} → ∞ A → A
#_ : {A : Set} → A → ∞ A
```

<sup>3</sup> At the time of writing this is not exactly true in Agda (Danielsson and Altenkirch 2009, Section 7.1), but the difference between *List A* and the fixpoint is irrelevant for the purposes of this paper. Similar considerations apply to greatest fixpoints.

The constructor  $\sharp$  is a tightly binding prefix operator. Ordinary function application binds tighter, though.

Values of coinductive types can be constructed using *guarded* corecursion (Coquand 1994):

$$\begin{aligned} \text{map}_S &: \{A\ B : \text{Set}\} \rightarrow (A \rightarrow B) \rightarrow \text{Stream } A \rightarrow \text{Stream } B \\ \text{map}_S f\ (x :: xs) &= f\ x :: \sharp\ \text{map}_S f\ (\flat\ xs) \end{aligned}$$

The definition of  $\text{map}_S$  is accepted by Agda’s termination checker because the corecursive call is guarded by  $\sharp$ , without any non-constructor function between the left-hand side and the corecursive call. This syntactic notion of guardedness ensures that corecursive definitions are *productive*: even if the value being constructed is infinite, the next constructor can always be computed in a finite number of steps.

### 2.3 Coinductive Relations

Let us now consider a coinductively defined *relation*: stream equality, or bisimilarity. Two streams are equal if they have identical heads and their tails are equal (coinductively):

$$\frac{\flat\ xs \approx \flat\ ys}{x :: xs \approx x :: ys} \quad (\text{coinductive})$$

This inference system can be represented using an indexed data type:

$$\begin{aligned} \mathbf{data}\ \_ \approx \_ &: \{A : \text{Set}\} : \text{Stream } A \rightarrow \text{Stream } A \rightarrow \text{Set} \mathbf{where} \\ \_ :: \_ &: (x : A) \{xs\ ys : \infty (\text{Stream } A)\} \rightarrow \infty (\flat\ xs \approx \flat\ ys) \rightarrow \\ &x :: xs \approx x :: ys \end{aligned}$$

Note the use of *dependent* function spaces  $((x : A) \rightarrow B$  where  $x$  can occur in  $B$ ) to set up dependencies of types on values. Note also that the elements of the type  $xs \approx ys$  are *proofs* witnessing the equality of  $xs$  and  $ys$ , and that the termination checker ensures that these proofs are productive.

Elements of coinductively defined relations can be constructed using corecursion. As an example, let us prove the map-iterate property (Gibbons and Hutton 2005):

$$\text{map}_S f\ (\text{iterate } f\ x) \approx \text{iterate } f\ (f\ x).$$

The function *iterate* repeatedly applies a function to a seed element and collects the results in a stream:

$$\text{iterate } f\ x = x :: \sharp\ (f\ x :: \sharp\ (f\ (f\ x) :: \dots)).$$

The function is defined corecursively:

$$\begin{aligned} \text{iterate} &: \{A : \text{Set}\} \rightarrow (A \rightarrow A) \rightarrow A \rightarrow \text{Stream } A \\ \text{iterate } f \ x &= x :: \# \text{iterate } f \ (f \ x) \end{aligned}$$

The map-iterate property can be proved by using guarded corecursion (the term *guarded coinduction* could also be used):

$$\begin{aligned} \text{map-iterate} &: \{A : \text{Set}\} (f : A \rightarrow A) (x : A) \rightarrow \\ &\quad \text{map}_S f \ (\text{iterate } f \ x) \approx \text{iterate } f \ (f \ x) \\ \text{map-iterate } f \ x &= f \ x :: \# \text{map-iterate } f \ (f \ x) \end{aligned}$$

To see how this proof works, consider how it can be built up step by step (as in an interactive Agda session):

$$\text{map-iterate } f \ x = ?$$

The type of the goal  $?$  is  $\text{map}_S f \ (\text{iterate } f \ x) \approx \text{iterate } f \ (f \ x)$ . Agda types should always be read up to normalisation, so this is equivalent to<sup>4</sup>

$$f \ x :: \# \text{map}_S f \ (\text{b} \ (\# \text{iterate } f \ (f \ x))) \approx f \ x :: \# \text{iterate } f \ (f \ (f \ x)).$$

(Note that normalisation does not involve reduction under  $\#$ , and that  $\text{b} \ (\# \ x)$  reduces to  $x$ .) This type matches the result type of the equality constructor  $_{::=}$ , so we can refine the goal:

$$\text{map-iterate } f \ x = f \ x :: ?$$

The new goal type is

$$\infty \ (\text{map}_S f \ (\text{iterate } f \ (f \ x)) \approx \text{iterate } f \ (f \ (f \ x))),$$

so the proof can be finished by an application of the coinductive hypothesis under the guarding constructor  $\#$ :

$$\text{map-iterate } f \ x = f \ x :: \# \text{map-iterate } f \ (f \ x)$$

## 2.4 Mixed Induction and Coinduction

The types above are either inductive or coinductive. Let us now consider a type which uses both induction and coinduction. Hancock et al. (2009) define a language of stream processors, representing functions of type  $\text{Stream } A \rightarrow \text{Stream } B$ , using a nested fixpoint:  $\nu Y. \mu X. B \times Y + (A \rightarrow X)$ . We can represent this fixpoint in Agda as follows:

```
data SP (A B : Set) : Set where
  put : B → ∞ (SP A B) → SP A B
  get : (A → SP A B) → SP A B
```

---

<sup>4</sup> This is a simplification of the current behaviour of Agda.

The stream processor `put  $b$   $sp$`  outputs  $b$ , and continues processing according to  $sp$ . The processor `get  $f$`  reads one element  $a$  from the input stream, and continues processing according to  $f a$ . In the case of `put` the recursive argument is coinductive, so it is fine to output an infinite number of elements, whereas in the case of `get` the recursive argument is inductive, which means that one can only read a finite number of elements before writing the next one. This ensures that the output stream can be generated productively.

We can implement a simple stream processor which copies the input to the output as follows:

$$\begin{aligned} copy &: \{A : Set\} \rightarrow SP\ A\ A \\ copy &= \text{get } (\lambda a \rightarrow \text{put } a\ (\# copy)) \end{aligned}$$

This definition is guarded. Note that `copy` contains an infinite number of `get` constructors. This is fine, even though `get`'s argument is inductive, because there is never a stretch of infinitely many `get` constructors without an intervening delay constructor (`#_`).

As another example we can compute the semantics of a stream processor:

$$\begin{aligned} \llbracket \_ \rrbracket &: \{A\ B : Set\} \rightarrow SP\ A\ B \rightarrow Stream\ A \rightarrow Stream\ B \\ \llbracket \text{put } b\ sp \rrbracket as &= b :: \# (\llbracket b\ sp \rrbracket as) \\ \llbracket \text{get } f \rrbracket (a :: as) &= \llbracket f\ a \rrbracket (\flat as) \end{aligned}$$

( $\llbracket \_ \rrbracket$  is a mixfix operator.) This definition uses a lexicographic combination of guarded corecursion and higher-order structural recursion. In the first clause the corecursive call is guarded. In the second clause it “preserves guardedness” (it takes place under zero coinductive constructors rather than one), and the first argument is structurally smaller.

## 2.5 Relations Using Mixed Induction and Coinduction

As a final example we define a relation using mixed induction and coinduction. Capretta (2005) defines the partiality monad, which can be used to represent potentially non-terminating computations, as follows:

$$\begin{aligned} \mathbf{data} \_{}^\nu (A : Set) &: Set \mathbf{where} \\ \mathbf{return} &: A \rightarrow A^\nu \\ \mathbf{step} &: \infty (A^\nu) \rightarrow A^\nu \end{aligned}$$

The constructor `return` returns a result, and `step` postpones a computation. Non-termination is represented as an infinitely postponed computation:

$$\begin{aligned} \perp &: \{A : Set\} \rightarrow A^\nu \\ \perp &= \mathbf{step} (\# \perp) \end{aligned}$$

A natural definition of equality for partial computations is weak bisimilarity (viewing `step` as a silent transition):<sup>5</sup>

<sup>5</sup> In order to reduce clutter the declarations of implicit arguments have been omitted in the remainder of the paper.

```

data  $\cong$  :  $A^\nu \rightarrow A^\nu \rightarrow \text{Set}$  where
  return :  $\text{return } v \cong \text{return } v$ 
  step :  $\infty (\text{b } x \cong \text{b } y) \rightarrow \text{step } x \cong \text{step } y$ 
  stepr :  $x \cong \text{b } y \rightarrow x \cong \text{step } y$ 
  stepl :  $\text{b } x \cong y \rightarrow \text{step } x \cong y$ 

```

This is basically the congruence generated by **return** and **step**, but allowing for finite differences in delay. Note that the requirement of *finite* differences in delay is captured by the use of induction for **step**<sup>r</sup> and **step**<sup>l</sup>, while the use of coinduction for **step** is necessary to be able to prove that the relation is reflexive.

### 3 Recursive Types

Brandt and Henglein (1998) define the following language of recursive types:

$$\sigma, \tau ::= \perp \mid \top \mid X \mid \sigma \rightarrow \tau \mid \mu X. \sigma \rightarrow \tau$$

Here  $\perp$  and  $\top$  are the least and greatest types, respectively,  $X$  is a variable,  $\sigma \rightarrow \tau$  is a function type, and  $\mu X. \sigma \rightarrow \tau$  is a fixpoint, with bound variable  $X$ . (The body of the fixpoint is required to be a function type, so types like  $\mu X. X$  are ruled out.) The intention is that a fixpoint  $\mu X. \sigma \rightarrow \tau$  should be equivalent to its unfolding  $(\sigma \rightarrow \tau)[X := \mu X. \sigma \rightarrow \tau]$ .

The language above can be represented in Agda as follows:

```

data Ty (n : ℕ) : Set where
  ⊥      : Ty n
  ⊤      : Ty n
  var    : Fin n → Ty n
   $\rightarrow$     : Ty n → Ty n → Ty n
   $\mu \rightarrow$  : Ty (1 + n) → Ty (1 + n) → Ty n

```

Here variables are represented using de Bruijn indices:  $Ty\ n$  represents types with at most  $n$  free variables, and  $Fin\ n$  is a type representing the first  $n$  natural numbers. Substitution can also be defined;  $\sigma \llbracket \tau \rrbracket$  is the capture-avoiding substitution of  $\tau$  for variable 0 in  $\sigma$ :

$$\llbracket - \rrbracket : Ty\ (1 + n) \rightarrow Ty\ n \rightarrow Ty\ n$$

The following function unfolds a fixpoint one step:

```

unfold< $\mu \rightarrow$ > : Ty (1 + n) → Ty (1 + n) → Ty n
unfold< $\mu \sigma \rightarrow \tau$ > = ( $\sigma \rightarrow \tau$ )  $\llbracket \mu \sigma \rightarrow \tau \rrbracket$ 

```

(Note that  $\mu \rightarrow$ ,  $\llbracket - \rrbracket$  and  $\text{unfold}\langle \mu \rightarrow \rangle$  are all infix operators which take two arguments.)

## 4 Subtyping via Trees

A natural definition of subtyping goes via subtyping for potentially infinite trees (Gapeyev et al. 2002):

```
data Tree (n : ℕ) : Set where
  ⊥      : Tree n
  ⊤      : Tree n
  var    : Fin n → Tree n
  ->-    : ∞ (Tree n) → ∞ (Tree n) → Tree n
```

The subtyping relation for trees can be given coinductively as follows:

```
data -≤Tree- : Tree n → Tree n → Set where
  ⊥      : ⊥      ≤Tree τ
  ⊤      : σ      ≤Tree ⊤
  var    : var x  ≤Tree var x
  ->-    : ∞ (⌊ τ1 ≤Tree ⌊ σ1 ⌋ → ∞ (⌊ σ2 ≤Tree ⌊ τ2 ⌋ →
    σ1 → σ2 ≤Tree τ1 → τ2
```

Note the contravariant treatment of the codomain of the function space.

The semantics of a recursive type can be given in terms of its unfolding as a potentially infinite tree:

```
⌊ - ⌋ : Ty n → Tree n
⌊ ⊥ ⌋      = ⊥
⌊ ⊤ ⌋      = ⊤
⌊ var x ⌋   = var x
⌊ σ → τ ⌋  = # ⌊ σ ⌋ → # ⌊ τ ⌋
⌊ μ σ → τ ⌋ = # ⌊ σ [ χ ] ⌋ → # ⌊ τ [ χ ] ⌋
               where χ = μ σ → τ
```

The subtyping relation for types can then be defined by combining  $-≤_{\text{Tree}}$  and  $\llbracket - \rrbracket$ :


```
-≤Type- : Ty n → Ty n → Set
σ ≤Type τ = ⌊ σ ⌋ ≤Tree ⌊ τ ⌋
```

As a simple example, consider the following two types,  $\sigma = \mu X. X \rightarrow X$  and  $\tau = \mu X. (X \rightarrow \perp) \rightarrow \top$ :

```
σ : Ty 0
σ = μ var zero → var zero
τ : Ty 0
τ = μ (var zero → ⊥) → ⊤
```

The first few levels of the infinite trees corresponding to these types can be seen in Fig. 1. It is straightforward to show that  $\sigma$  is a subtype of  $\tau$  using a corecursive proof:





**Fig. 1.** The first levels of the infinite trees corresponding to the types  $\mu X. X \rightarrow X$  and  $\mu X. (X \rightarrow \perp) \rightarrow \top$ .

$$\begin{aligned} \sigma \leq \tau &: \sigma \leq_{\text{Type}} \tau \\ \sigma \leq \tau &= \# (\# \sigma \leq \tau \rightarrow \# \perp) \rightarrow \# \top \end{aligned}$$

(Note that  $\sigma \leq \tau$  is an identifier.)

Amadio and Cardelli (1993) also define subtyping for recursive types by going via potentially infinite trees, but they define a subtyping relation *inductively* on finite trees, and state that an infinite tree  $\sigma$  is a subtype of another tree  $\tau$  when every finite approximation (of a certain kind) of  $\sigma$  is a subtype of the corresponding approximation of  $\tau$ . It is easy to show that this definition, as adapted by Brandt and Henglein (1998), is equivalent to the one given above. One direction of the proof uses induction on the depth of approximation, and the other constructs elements of  $\sigma \leq_{\text{Type}} \tau$  corecursively; see the code which accompanies the paper (Danielsson 2009).

## 5 Subtyping Using Mixed Induction and Coinduction

Subtyping can also be defined directly, without going via trees. The following definition is inspired by one given by Brandt and Henglein (1998), see Sect. 6:

```

data  $\_ \leq \_$  :  $Ty\ n \rightarrow Ty\ n \rightarrow Set$  where
   $\perp$       :  $\perp \leq \tau$ 
   $\top$       :  $\sigma \leq \top$ 
   $\_ \rightarrow \_$  :  $\infty (\tau_1 \leq \sigma_1) \rightarrow \infty (\sigma_2 \leq \tau_2) \rightarrow \sigma_1 \rightarrow \sigma_2 \leq \tau_1 \rightarrow \tau_2$ 
  unfold :  $\mu \tau_1 \rightarrow \tau_2 \leq \text{unfold}(\mu \tau_1 \rightarrow \tau_2)$ 
  fold   :  $\text{unfold}(\mu \tau_1 \rightarrow \tau_2) \leq \mu \tau_1 \rightarrow \tau_2$ 
  refl   :  $\tau \leq \tau$ 
  trans  :  $\tau_1 \leq \tau_2 \rightarrow \tau_2 \leq \tau_3 \rightarrow \tau_1 \leq \tau_3$ 

```

Note that the structural rules ( $\perp$ ,  $\top$ ,  $\_ \rightarrow \_$ ) are defined coinductively, while the other rules, most importantly **trans**, are defined inductively.

Now, if we can prove that the relation  $\_ \leq \_$  is equivalent to  $\_ \leq_{\text{Type}} \_$  (and thus also equivalent to Amadio and Cardelli's relation), then we have showed what we

set out to show: that coinduction and the rule of transitivity can be combined. We can prove completeness by a simple application of guarded corecursion (omitted here):

$$complete : \sigma \leq_{\text{Type}} \tau \rightarrow \sigma \leq \tau$$

The soundness proof is a little more tricky. The following lemmas are easy to prove:

$$\begin{aligned} unfold_{\text{Type}} &: \mu \tau_1 \rightarrow \tau_2 \leq_{\text{Type}} unfold \langle \mu \tau_1 \rightarrow \tau_2 \rangle \\ fold_{\text{Type}} &: unfold \langle \mu \tau_1 \rightarrow \tau_2 \rangle \leq_{\text{Type}} \mu \tau_1 \rightarrow \tau_2 \\ refl_{\text{Type}} &: \tau \leq_{\text{Type}} \tau \\ trans_{\text{Type}} &: \tau_1 \leq_{\text{Type}} \tau_2 \rightarrow \tau_2 \leq_{\text{Type}} \tau_3 \rightarrow \tau_1 \leq_{\text{Type}} \tau_3 \end{aligned}$$

Using these lemmas one might think that the following should be accepted as a soundness proof:

$$\begin{aligned} sound &: \sigma \leq \tau \rightarrow \sigma \leq_{\text{Type}} \tau \\ sound \perp &= \perp \\ sound \top &= \top \\ sound (\tau_1 \leq \sigma_1 \rightarrow \sigma_2 \leq \tau_2) &= \# sound (\text{b } \tau_1 \leq \sigma_1) \rightarrow \# sound (\text{b } \sigma_2 \leq \tau_2) \\ sound unfold &= unfold_{\text{Type}} \\ sound fold &= fold_{\text{Type}} \\ sound refl &= refl_{\text{Type}} \\ sound (trans \tau_1 \leq \tau_2 \tau_2 \leq \tau_3) &= trans_{\text{Type}} (sound \tau_1 \leq \tau_2) (sound \tau_2 \leq \tau_3) \end{aligned}$$

However, consider the case for **trans**. The arguments to the recursive calls are structurally smaller than the inputs, but  $trans_{\text{Type}}$  is not a constructor, so guardedness is not preserved. The proof is productive (given a suitable definition of  $trans_{\text{Type}}$ ), but Agda's termination checker cannot see this.

In the absence of improved termination checking for Agda we provide a workaround. If  $trans_{\text{Type}}$  had been a constructor then the definition of *sound* would have been accepted, and this observation can be used to rescue the proof. First we define a variant of  $\leq_{\text{Tree}}$  which includes an extra inductive constructor, **trans**:

$$\begin{aligned} \text{data } \leq_{\text{TreeP}} &: \text{Tree } n \rightarrow \text{Tree } n \rightarrow \text{Set where} \\ \perp &: \perp \leq_{\text{TreeP}} \tau \\ \top &: \sigma \leq_{\text{TreeP}} \top \\ \text{var} &: \text{var } x \leq_{\text{TreeP}} \text{var } x \\ \rightarrow &: \infty (\text{b } \tau_1 \leq_{\text{TreeP}} \text{b } \sigma_1) \rightarrow \infty (\text{b } \sigma_2 \leq_{\text{TreeP}} \text{b } \tau_2) \rightarrow \\ &\quad \sigma_1 \rightarrow \sigma_2 \leq_{\text{TreeP}} \tau_1 \rightarrow \tau_2 \\ \text{trans} &: \tau_1 \leq_{\text{TreeP}} \tau_2 \rightarrow \tau_2 \leq_{\text{TreeP}} \tau_3 \rightarrow \tau_1 \leq_{\text{TreeP}} \tau_3 \end{aligned}$$

The letter P stands for “program”; this type defines a small language of equality

proof programs. It is easy to turn proofs into proof programs corecursively:

$$\ulcorner \_ \urcorner : \sigma \leq_{\text{Tree}} \tau \rightarrow \sigma \leq_{\text{TreeP}} \tau$$

We can now write a guarded proof *program* which “proves” soundness:

$$\begin{aligned} \text{sound}_P : \sigma \leq \tau &\rightarrow \llbracket \sigma \rrbracket \leq_{\text{TreeP}} \llbracket \tau \rrbracket \\ \text{sound}_P \perp &= \perp \\ \text{sound}_P \top &= \top \\ \text{sound}_P (\tau_1 \leq \sigma_1 \rightarrow \sigma_2 \leq \tau_2) &= \# \text{sound}_P (\text{b } \tau_1 \leq \sigma_1) \rightarrow \# \text{sound}_P (\text{b } \sigma_2 \leq \tau_2) \\ \text{sound}_P \text{unfold} &= \ulcorner \text{unfold}_{\text{Type}} \urcorner \\ \text{sound}_P \text{fold} &= \ulcorner \text{fold}_{\text{Type}} \urcorner \\ \text{sound}_P \text{refl} &= \ulcorner \text{refl}_{\text{Type}} \urcorner \\ \text{sound}_P (\text{trans } \tau_1 \leq \tau_2 \ \tau_2 \leq \tau_3) &= \text{trans } (\text{sound}_P \ \tau_1 \leq \tau_2) (\text{sound}_P \ \tau_2 \leq \tau_3) \end{aligned}$$

If we can also find a way to turn proof programs into proofs, productively, then we are done. We start by defining a type of weak head normal forms (WHNFs) for the proof programs:

$$\begin{aligned} \text{data } \_ \leq_{\text{TreeW}} : \text{Tree } n &\rightarrow \text{Tree } n \rightarrow \text{Set where} \\ \perp &: \perp \leq_{\text{TreeW}} \tau \\ \top &: \sigma \leq_{\text{TreeW}} \top \\ \text{var} &: \text{var } x \leq_{\text{TreeW}} \text{var } x \\ \_ \rightarrow \_ &: \text{b } \tau_1 \leq_{\text{TreeP}} \text{b } \sigma_1 \rightarrow \text{b } \sigma_2 \leq_{\text{TreeP}} \text{b } \tau_2 \rightarrow \\ &\sigma_1 \rightarrow \sigma_2 \leq_{\text{TreeW}} \tau_1 \rightarrow \tau_2 \end{aligned}$$

Note that the arguments to  $\_ \rightarrow \_$  are *programs*, not WHNFs. One can prove by simple case analysis that  $\_ \leq_{\text{TreeW}}$  is transitive:

$$\text{trans}_{\text{TreeW}} : \tau_1 \leq_{\text{TreeW}} \tau_2 \rightarrow \tau_2 \leq_{\text{TreeW}} \tau_3 \rightarrow \tau_1 \leq_{\text{TreeW}} \tau_3$$

From this result it follows by structural recursion that programs can be turned into WHNFs:

$$\begin{aligned} \text{whnf} : \sigma \leq_{\text{TreeP}} \tau &\rightarrow \sigma \leq_{\text{TreeW}} \tau \\ \text{whnf } \perp &= \perp \\ \text{whnf } \top &= \top \\ \text{whnf } \text{var} &= \text{var} \\ \text{whnf } (\tau_1 \leq \sigma_1 \rightarrow \sigma_2 \leq \tau_2) &= \text{b } \tau_1 \leq \sigma_1 \rightarrow \text{b } \sigma_2 \leq \tau_2 \\ \text{whnf } (\text{trans } \tau_1 \leq \tau_2 \ \tau_2 \leq \tau_3) &= \text{trans}_{\text{TreeW}} (\text{whnf } \tau_1 \leq \tau_2) (\text{whnf } \tau_2 \leq \tau_3) \end{aligned}$$

By using the *whnf* function repeatedly one can then turn proof programs into “actual” proofs:

**mutual**

$$\begin{aligned}
\llbracket - \rrbracket_W &: \sigma \leq_{\text{TreeW}} \tau \rightarrow \sigma \leq_{\text{Tree}} \tau \\
\llbracket \perp \rrbracket_W &= \perp \\
\llbracket \top \rrbracket_W &= \top \\
\llbracket \text{var} \rrbracket_W &= \text{var} \\
\llbracket \tau_1 \leq \sigma_1 \rightarrow \sigma_2 \leq \tau_2 \rrbracket_W &= \# \llbracket \tau_1 \leq \sigma_1 \rrbracket_P \rightarrow \# \llbracket \sigma_2 \leq \tau_2 \rrbracket_P \\
\llbracket - \rrbracket_P &: \sigma \leq_{\text{TreeP}} \tau \rightarrow \sigma \leq_{\text{Tree}} \tau \\
\llbracket \sigma \leq \tau \rrbracket_P &= \llbracket \text{whnf } \sigma \leq \tau \rrbracket_W
\end{aligned}$$

Note that these functions are guarded and hence productive. Finally we get the soundness proof:

$$\begin{aligned}
\text{sound} &: \sigma \leq \tau \rightarrow \sigma \leq_{\text{Type}} \tau \\
\text{sound } \sigma \leq \tau &= \llbracket \text{sound}_P \sigma \leq \tau \rrbracket_P
\end{aligned}$$

## 6 Inductive Axiomatisation of Subtyping

Brandt and Henglein (1998) do not define subtyping using mixed induction and coinduction, as in Sect. 5, but using an inductive *encoding* of coinduction. Their subtyping relation is ternary:  $A \vdash \sigma \leq \tau$  means that  $\sigma$  is a subtype of  $\tau$  given the assumptions in  $A$ . An assumption (a hypothesis) is simply a pair of types:

$$\begin{aligned}
\text{data Hyp } (n : \mathbb{N}) &: \text{Set where} \\
\sim &: \text{Ty } n \rightarrow \text{Ty } n \rightarrow \text{Hyp } n
\end{aligned}$$

The subtyping relation is defined as follows:

$$\begin{aligned}
\text{data } \_ \vdash \_ \leq \_ & (A : \text{List } (\text{Hyp } n)) : \text{Ty } n \rightarrow \text{Ty } n \rightarrow \text{Set where} \\
\perp &: A \vdash \perp \leq \tau \\
\top &: A \vdash \sigma \leq \top \\
\rightarrow &: \text{let } H = \sigma_1 \rightarrow \sigma_2 \sim \tau_1 \rightarrow \tau_2 \text{ in} \\
&H :: A \vdash \tau_1 \leq \sigma_1 \rightarrow H :: A \vdash \sigma_2 \leq \tau_2 \rightarrow \\
&A \vdash \sigma_1 \rightarrow \sigma_2 \leq \tau_1 \rightarrow \tau_2 \\
\text{unfold} &: A \vdash \mu \tau_1 \rightarrow \tau_2 \leq \text{unfold} \langle \mu \tau_1 \rightarrow \tau_2 \rangle \\
\text{fold} &: A \vdash \text{unfold} \langle \mu \tau_1 \rightarrow \tau_2 \rangle \leq \mu \tau_1 \rightarrow \tau_2 \\
\text{refl} &: A \vdash \tau \leq \tau \\
\text{trans} &: A \vdash \tau_1 \leq \tau_2 \rightarrow A \vdash \tau_2 \leq \tau_3 \rightarrow A \vdash \tau_1 \leq \tau_3 \\
\text{hyp} &: \sigma \sim \tau \in A \rightarrow A \vdash \sigma \leq \tau
\end{aligned}$$

Here  $\_ \in \_$  encodes list membership. Note that coinduction is encoded in the  $\rightarrow$  rule by inclusion of the consequent in the lists of assumptions of the antecedents.

Brandt and Henglein prove that their relation (with an empty list of assumptions) is equivalent to Amadio and Cardelli's. Their proof is considerably more

complicated than the proof outlined above which shows that  $\_ \leq \_$  is equivalent to Amadio and Cardelli's definition, but as part of the proof they show that subtyping is decidable. By composing the two equivalence proofs we get that subtyping as defined in Sect. 5 is also decidable.

Brandt and Henglein use a classical argument to show that their algorithm terminates, so it is not entirely obvious that it can be implemented in a total, constructive type theory like Agda. However, we have adapted the algorithm to this setting:

$$\_ \leq? \_ : (\sigma \tau : Ty\ n) \rightarrow Dec ([\ ] \vdash \sigma \leq \tau)$$

A value in  $Dec\ A$  is either a value in  $A$ , or a proof showing that no such value exists, so this decision procedure does not merely say “yes” or “no”, it backs up its verdict with solid evidence. Details of the implementation of  $\_ \leq? \_$  are available in the code accompanying the paper (Danielsson 2009).

We know that  $\_ \vdash \_ \leq \_$  is equivalent to  $\_ \leq \_$ , because both relations are equivalent to Amadio and Cardelli's. However, it can still be instructive to see a direct proof of soundness of  $\_ \vdash \_ \leq \_$  with respect to  $\_ \leq \_$ . The proof below uses a cyclic (but productive) proof to turn the inductive encoding of coinduction used in  $\_ \vdash \_ \leq \_$  into the “actual” coinduction used in  $\_ \leq \_$ .

To state soundness the type  $All$  is used;  $All\ P\ xs$  means that all elements in  $xs$  satisfy  $P$ :

$$\begin{aligned} \mathbf{data}\ All\ (P : A \rightarrow Set) : List\ A \rightarrow Set\ \mathbf{where} \\ [\ ] &: All\ P\ [\ ] \\ \_ :: \_ &: P\ x \rightarrow All\ P\ xs \rightarrow All\ P\ (x :: xs) \end{aligned}$$

The soundness proof shows that if  $A \vdash \sigma \leq \tau$ , where all pairs  $\sigma' \lesssim \tau'$  in  $A$  satisfy  $\sigma' \leq \tau'$ , then  $\sigma \leq \tau$ :

$$\begin{aligned} Valid &: (Ty\ n \rightarrow Ty\ n \rightarrow Set) \rightarrow Hyp\ n \rightarrow Set \\ Valid\ \_ R \_ &(\sigma_1 \lesssim \sigma_2) = \sigma_1\ R\ \sigma_2 \\ sound &: All\ (Valid\ \_ \leq \_) A \rightarrow A \vdash \sigma \leq \tau \rightarrow \sigma \leq \tau \end{aligned}$$

The interesting cases of *sound* are the ones for *hyp* and  $\_ \rightarrow \_$ . Hypotheses can simply be looked up in the list of valid assumptions (using *lookup* :  $All\ P\ xs \rightarrow x \in xs \rightarrow P\ x$ ), and function spaces can be handled by defining a cyclic proof:

$$\begin{aligned} sound\ valid\ (\mathbf{hyp}\ h) &= lookup\ valid\ h \\ sound\ valid\ (\tau_1 \leq \sigma_1 \rightarrow \sigma_2 \leq \tau_2) &= proof \\ \mathbf{where}\ proof &= \# sound\ (proof :: valid)\ \tau_1 \leq \sigma_1 \rightarrow \\ &\# sound\ (proof :: valid)\ \sigma_2 \leq \tau_2 \end{aligned}$$

Note that the corecursive calls extend the list of valid assumptions with the proof currently being defined.

The definition of *proof* above is not guarded, but it would be if *sound* were a constructor. We use the technique from Sect. 5 to make the proof guarded. The program and WHNF types can be defined mutually as follows:

### mutual

```

data  $\_ \leq_P \_ : Ty\ n \rightarrow Ty\ n \rightarrow Set$  where
  sound :  $All\ (Valid\ \_ \leq_W \_) A \rightarrow A \vdash \sigma \leq \tau \rightarrow \sigma \leq_P \tau$ 
data  $\_ \leq_W \_ : Ty\ n \rightarrow Ty\ n \rightarrow Set$  where
  done :  $\sigma \leq \tau \rightarrow \sigma \leq_W \tau$ 
   $\_ \rightarrow \_ : \infty\ (\tau_1 \leq_P \sigma_1) \rightarrow \infty\ (\sigma_2 \leq_P \tau_2) \rightarrow \sigma_1 \rightarrow \sigma_2 \leq_W \tau_1 \rightarrow \tau_2$ 
  trans :  $\tau_1 \leq_W \tau_2 \rightarrow \tau_2 \leq_W \tau_3 \rightarrow \tau_1 \leq_W \tau_3$ 

```

The cases of *sound* listed above are now part of the *whnf* function:

```

whnf :  $\sigma \leq_P \tau \rightarrow \sigma \leq_W \tau$ 
...
whnf (sound valid (hyp h))           = lookup valid h
whnf (sound valid ( $\tau_1 \leq \sigma_1 \rightarrow \sigma_2 \leq \tau_2$ )) = proof
  where proof = # sound (proof :: valid)  $\tau_1 \leq \sigma_1 \rightarrow$ 
               # sound (proof :: valid)  $\sigma_2 \leq \tau_2$ 

```

Note that *proof* is now guarded.

We have not found a proof of completeness of  $\_ \vdash \_ \leq \_$  with respect to  $\_ \leq \_$  which does not use a decision procedure for subtyping. This is not entirely surprising; such a completeness proof must turn a potentially infinite proof of  $\sigma \leq \tau$  into a finite proof of  $[] \vdash \sigma \leq \tau$ , so some “trick” is necessary. With a suitably formulated decision procedure at hand the trick is simple. We have implemented a decision procedure which gives either a proof of  $[] \vdash \sigma \leq \tau$ , or a proof which shows that  $\sigma \leq \tau$  is impossible. In the first case we are done, and in the second case a contradiction can be derived.

## 7 Postulating an Admissible Rule May Not Be Sound

Given an inductively defined inference system one can add a new rule corresponding to an admissible property without changing the set of derivable properties. It is easy to prove this statement by defining functions which translate between the two inference systems. Translating derivations from the old to the new inference system is trivial. When translating in the other direction one can replace all occurrences of the new rule with instances of the proof of admissibility; this process can be implemented using recursion over the structure of the input derivation.

However, when coinduction comes into the picture this property no longer holds (de Vries 2009). The proof given above breaks down because there is no guarantee that the second translation can be implemented in a productive way. The problem is that, although the admissible rule has a proof, this proof may not be sufficiently “contractive” (for instance, the proof may replace coinductive rules in the input derivation with inductive rules in the output derivation).

The following example illustrates the problem. Recall the definition of the partiality monad in Sect. 2.5. One can prove that the equality  $\_ \cong \_$  is an equivalence relation, and that it is not trivial (assuming that the result type  $A$  is inhabited). Let us now add transitivity as an inductive rule:

**data**  $\cong$  :  $A^\nu \rightarrow A^\nu \rightarrow \text{Set}$  **where**

...

**trans** :  $x \cong y \rightarrow y \cong z \rightarrow x \cong z$

Given this new constructor we can prove, using guarded coinduction, that the relation is trivial:

$$\begin{aligned} \text{trivial} &: (x\ y : A^\nu) \rightarrow x \cong y \\ \text{trivial } x\ y &= \text{trans} (\text{step}^r (\text{refl } x)) \\ &\quad (\text{trans} (\text{step} (\sharp \text{trivial } x\ y)) \\ &\quad (\text{step}^l (\text{refl } y))) \end{aligned}$$

(Here *refl* is a proof of reflexivity.)

This problem does not affect the definition of subtyping given above, which has been proved to be equivalent to other definitions from the literature. However, it means that one should exercise caution when defining relations using mixed induction and coinduction, and avoid relying on results or intuitions which are only valid in the inductive case.

There are actually several different ways in which one can close a coinductively defined binary relation

**data**  $\sim$  :  $A \rightarrow A \rightarrow \text{Set}$  **where**

...

under transitivity. We list three:

1. One can include transitivity as a coinductive constructor:

**data**  $\sim$  :  $A \rightarrow A \rightarrow \text{Set}$  **where**

...

**trans** :  $x \sim y \rightarrow y \sim z \rightarrow x \sim z$

This amounts to defining the largest relation which is closed under transitivity, and is not very useful, as pointed out in the introduction.

2. One can define the least relation which includes  $\sim$  and is closed under transitivity:

**data**  $\sim'$  :  $A \rightarrow A \rightarrow \text{Set}$  **where**

**include** :  $x \sim y \rightarrow x \sim' y$

**trans** :  $x \sim' y \rightarrow y \sim' z \rightarrow x \sim' z$

This “solves” the problem outlined above, because if  $\sim$  is transitive, then  $\sim$  and  $\sim'$  are equivalent. However, in any given proof **trans** can only be used a finite number of times, and this can be a rather severe restriction. For instance, the definition of  $\leq$  in Sect. 5 would not have been correct if transitivity had been included using this method.

3. Finally one can include transitivity as an inductive constructor, like in previous sections:

```

data  $\sim$  :  $A \rightarrow A \rightarrow Set$  where
  ...
  trans :  $x \sim y \rightarrow y \sim z \rightarrow x \sim z$ 

```

This definition often gives a more useful notion of transitivity than the one above, because transitivity can be used anywhere in a proof, infinitely often, as long as there is never a stretch of infinitely many transitivity constructors without any intervening coinductive constructor. However, this notion of transitivity can sometimes be too strong, as illustrated for the partiality monad equality  $\cong$  above: the “infinitely transitive closure” is sometimes the trivial relation.

## 8 Conclusions

We have showed that coinduction can be usefully combined with the rule of transitivity, and discussed under what conditions the technique is applicable. We have also defined subtyping for recursive types in a new way, and compared this definition to a similar axiomatisation given by Brandt and Henglein (1998). Brandt and Henglein note that their inductive encoding of coinduction seems to be closely related to guarded coinduction, but leave a precise comparison to future work. This paper provides a precise comparison, albeit not for the general case, but only for a particular example (the subtyping relations given in Sects. 5 and 6).

It is our hope that this paper provides a compelling example of the use of mixed induction and coinduction. We have found this technique useful in a number of situations, and encourage more programming language researchers—as well as programmers interested in guaranteed totality—to become familiar with it.

## Acknowledgements

Nils Anders Danielsson would like to thank Peter Hancock for introducing him to the technique of mixed induction and coinduction. Thorsten Altenkirch would like to thank Nicolas Oury for joint work on  $II\Sigma$  which has had an impact on the work in this paper, and Graham Hutton for earlier joint but unpublished work on coinductive reasoning. Both authors would like to thank Conor McBride, Nicolas Oury and Anton Setzer for many discussions about coinduction which have influenced this work, and Graham Hutton and Henrik Nilsson for feedback which improved the presentation. Both authors gratefully acknowledge funding from EPSRC.



## References

- Andreas Abel. Mixed inductive/coinductive types and strong normalization. In *Programming Languages and Systems, 5th Asian Symposium, APLAS 2007*, volume 4807 of *LNCS*, pages 286–301, 2009.
- The Agda Team. The Agda Wiki. Available at <http://wiki.portal.chalmers.se/agda/>, 2009.
- Roberto M. Amadio and Luca Cardelli. Subtyping recursive types. *ACM Transactions on Programming Languages and Systems*, 15(4):575–631, 1993.
- G. Barthe, M. J. Frade, E. Giménez, L. Pinto, and T. Uustalu. Type-based termination of recursive definitions. *Mathematical Structures in Computer Science*, 14(1):97–141, 2004.
- Jon Barwise. *The Situation in Logic*, volume 17 of *CSLI Lecture Notes*, chapter Mixed Fixed Points. Center for the Study of Language and Information, Leland Stanford Junior University, 1989.
- Julian Bradfield and Colin Stirling. Modal mu-calculi. In *Handbook of Modal Logic*, volume 3 of *Studies in Logic and Practical Reasoning*. Elsevier, 2007.
- Michael Brandt and Fritz Henglein. Coinductive axiomatization of recursive type equality and subtyping. *Fundamenta Informaticae*, 33(4):309–338, 1998.
- Venanzio Capretta. General recursion via coinductive types. *Logical Methods in Computer Science*, 1(2):1–28, 2005.
- Thierry Coquand. Infinite objects in type theory. In *Types for Proofs and Programs, International Workshop TYPES’93*, volume 806 of *LNCS*, pages 62–78, 1994.
- Patrick Cousot and Radhia Cousot. Inductive definitions, semantics and abstract interpretations. In *POPL ’92, Proceedings of the 19th ACM SIGPLAN-SIGACT symposium on Principles of programming languages*, pages 83–94, 1992.
- Nils Anders Danielsson. Code accompanying the paper. Currently available from <http://www.cs.nott.ac.uk/~nad/>, 2009.
- Nils Anders Danielsson and Thorsten Altenkirch. Mixing induction and coinduction. Draft, 2009.
- Edsko de Vries. Re: [Coq-Club] Adding (inductive) transitivity to weak bisimilarity not sound? (was: Need help with coinductive proof). Message to the Coq-Club mailing list, August 2009.
- Vladimir Gapeyev, Michael Y. Levin, and Benjamin C. Pierce. Recursive subtyping revealed. *Journal of Functional Programming*, 12(6):511–548, 2002.
- Jeremy Gibbons and Graham Hutton. Proof methods for corecursive programs. *Fundamenta Informaticae*, 66(4):353–366, 2005.
- Eduardo Giménez. *Un Calcul de Constructions Infinies et son Application à la Vérification de Systèmes Communicants*. PhD thesis, Ecole Normale Supérieure de Lyon, 1996.
- Andrew D. Gordon. Bisimilarity as a theory of functional programming. *Theoretical Computer Science*, 228(1–2):5–47, 1999.
- Tatsuya Hagino. *A Categorical Programming Language*. PhD thesis, University of Edinburgh, 1987.
- Peter Hancock, Dirk Pattinson, and Neil Ghani. Representations of stream processors using nested fixed points. *Logical Methods in Computer Science*, 5(3:9), 2009.
- Ulrich Hensel and Bart Jacobs. Proof principles for datatypes with iterated recursion. In *Category Theory and Computer Science, 7th International Conference, CTCS ’97*, volume 1290 of *LNCS*, pages 220–241, 1997.

- John Hughes and Andrew Moran. Making choices lazily. In *FPCA '95, Proceedings of the seventh international conference on Functional programming languages and computer architecture*, pages 108–119, 1995.
- Dexter Kozen, Jens Palsberg, and Michael I. Schwartzbach. Efficient recursive subtyping. *Mathematical Structures in Computer Science*, 5(1):113–125, 1995.
- Xavier Leroy and Hervé Grall. Coinductive big-step operational semantics. *Information and Computation*, 207(2):284–304, 2009.
- Paul Blain Levy. Infinitary Howe’s method. In *Proceedings of the Eighth Workshop on Coalgebraic Methods in Computer Science (CMCS 2006)*, volume 164 of *ENTCS*, pages 85–104, 2006.
- Paul Francis Mendler. *Inductive Definition in Type Theory*. PhD thesis, Cornell University, 1988.
- Robin Milner. Operational and algebraic semantics of concurrent processes. In *Handbook of Theoretical Computer Science, Volume B: Formal Models and Semantics*. The MIT Press and Elsevier, 1990.
- Robin Milner and Mads Tofte. Co-induction in relational semantics. *Theoretical Computer Science*, 87(1):209–220, 1991.
- Olaf Müller, Tobias Nipkow, David von Oheimb, and Oscar Slotosch. HOLCF = HOL + LCF. *Journal of Functional Programming*, 9(2):191–223, 1999.
- Keiko Nakata and Tarmo Uustalu. Trace-based coinductive operational semantics for While; Big-step and small-step, relational and functional styles. In *Theorem Proving in Higher Order Logics, 22nd International Conference, TPHOLs 2009*, volume 5674 of *LNCS*, pages 375–390, 2009.
- Ulf Norell. *Towards a practical programming language based on dependent type theory*. PhD thesis, Chalmers University of Technology and Göteborg University, 2007.
- David Park. On the semantics of fair parallelism. In *Abstract Software Specifications*, volume 86 of *LNCS*, pages 504–526, 1980.
- Christophe Raffalli. *L’Arithmétique Fonctionnelle du Second Ordre avec Points Fixes*. PhD thesis, Université Paris VII, 1994.
- D. A. Turner. Total functional programming. *Journal of Universal Computer Science*, 10(7):751–768, 2004.
- Philip Wadler, Walid Taha, and David MacQueen. How to add laziness to a strict language, without even being odd. In *Proceedings of the 1998 ACM SIGPLAN Workshop on ML*, 1998.